United States
Environmental Protection
Agency

Office of Air Quality
Planning and Standards
Research Triangle Park NC 27711

EPA-450/4-81-022
May 1981

Air

EPA

USER'S MANUAL FOR MIXING HEIGHT COMPUTER PROGRAM

USER'S MANUAL FOR MIXING HEIGHT COMPUTER PROGRAM

Robert F. Kelly
Air Management Technology Branch
Monitoring and Data Analysis Division

U.S. ENVIRONMENTAL PROTECTION AGENCY
Office of Air, Noise and Radiation
Office of Air Quality Planning and Standards
Research Triangle Park, North Carolina 27711

May 1981

This report is issued by the Environmental Protection Agency to report technical data of interest to a limited number of readers. Copies are available free of charge to Federal employees, current contractors and grantees, and nonprofit organizations - in limited quantities - from the Library Services Office (MD-35), Research Triangle Park, North Carolina 27711; or, for a fee, from the National Technical Information Service, 5285 Port Royal Road, Springfield, Virginia 22161.

Publication No. EPA-450/4-81-022

TABLE OF CONTENTS

ABSTRACT

A FORTRAN-language computer program has been developed to estimate mixing height values for use in the Empirical Kinetic Modeling Approach/Ozone Isopleth Plotting Package (EKMA/OZIPP). This program uses temperature, pressure and height values measured at a surface site, and from atmospheric vertical profiles (e.g., radiosondes). The manual contains instructions on how to use the program, including a description of the data needed, how to format the data, and an explanation of the output from the program. Examples of input and output are also included.

LIST OF FIGURES

LIST OF TABLES

ACKNOWLEDGMENTS

Helpful suggestions for this report were given by Dr. Edwin L. Meyer and Mr. Norman C. Possiel. Messrs. Warren P. Freas and William N. Hamilton helped the author test, debug and punch copies of the computer program. The User's Manual was typed by Mrs. Carole J. Mask.

1.0 INTRODUCTION

This FORTRAN language computer program has been developed specifically to estimate mixing height values needed by the Empirical Kinetic Modeling Approach/ Ozone Isopleth Plotting Package (EKMA/OZIPP). The program was written in standard FORTRAN and tested on a UNIVAC 1100 computer system. Urban temperature and pressure values and data from a nearby upper air sounding site are used to estimate mixing height values for the urban area. This program enables non-meteorologists to estimate mixing heights from available data without using complex graphs or methods. The program statements are provided in Section 7.0 and a punched-card deck of the program may be obtained by writing or calling: Mr. Robert F. Kelly, U.S. Environmental Protection Agency, Office of Air Quality Planning and Standards, Mail Drop 14, Research Triangle Park, North Carolina 27711, (919) 541-5522.

2.0 METHOD USED

Mixing heights can be found by following a series of mathematical steps (like the example in Figure 1) or from a graph of temperature plotted against height or pressure (such as the graph in Figure 2). The mathematical method can be a long, tedious procedure. The graphical method, while easier to do, is less precise.

This program applies the mathematical version of the graphical method and is based on the steps in Figure 1. Measurements of temperature and pressure at the surface and at levels aloft are used to calculate values of potential temperature with height, as needed by the procedure. Potential temperature is a meteorological

Step 1 -- For reference, the information at the top of Table A-4 should be listed (e.g., date, city, etc.). If the morning mixing height is to be calculated, the 0800 LCT surface data are used. If the maximum mixing height is to be calculated, the data corresponding to the time of maximum temperature (i.e., between 800-1800 LCT) are used. In the row labeled URBAN SURFACE DATA, enter the following information: 1) the elevation of the urban temperature site in meters above sea level; 2) the surface pressure in millibars (this value is P_{sfc}); and 3) the surface temperature in degrees Celsus (°C).

Convert the surface temperature in column four to degrees Kelvin (°K) by adding 273.2, and enter the result in column five. This value is T_{sfc}(°K).

Use Equation 1 below and the values just entered to calculate the potential temperature at the surface (θ_{sfc} in °K to the nearest 0.1°K) and enter this value under column six "θ(°K)".

$$\theta_{sfc} \text{ (in °K)} = T_{sfc} \text{ (in °K)} \left(\frac{P_{sfc} \text{ (in mb)}}{1000 \text{ mb}} \right)^{-0.286} \qquad (1)$$

Step 2 -- Using the temperature sounding data, find the highest pressure level other than the sounding's surface value that is less than the pressure at the urban surface.* From this pressure level on the sounding, enter the height (if listed), pressure and temperature (in °C) into the row marked "(2)" on Table A-4.

Step 3 -- Convert the temperature at this level to the Kelvin scale and enter in column 5. Compute the potential temperature (θ_p) to the nearest 0.1°K using the pressure (P, in mb) and temperature (T_p in °K) at this level in Equation 2 below:

$$\theta_p \text{ (in °K)} = T_p \text{ (in °K)} \left(\frac{P \text{ (in mb)}}{1000 \text{ mb}} \right)^{-0.286} \qquad (2)$$

Enter the value of θ_p found from Equation (2) into the same row under the column labeled "θ(°K)".

Step 4 -- If the potential temperature, "θ," of the last row that was entered is greater than the potential temperature θ_{sfc}, and this is the first level above the surface, then 250 meters should be used as the mixing height. Otherwise, go to Step 5. If it is less than or equal to θ_{sfc}, then enter the height (if given), pressure and temperature of the next lowest pressure level found on the sounding into the next row of Table A-4 and return to Step 3.

Step 5 -- The mixing height is between the last two levels entered into Table A-4. If height values are given for both of these levels, the elevation of the mixing height can be found using Step 6. If one of the levels does not have a height value, use linear interpolation to find the pressure value for the potential temperature value of θ_{sfc} + 0.1°K. Enter this pressure value into the row marked "MIXING HEIGHT" at the bottom of Table A-4 under the column "PRESSURE in mb." Proceed to Step 7.

Step 6 -- From the two levels where height is given on the sounding surrounding the mixing height level, use linear interpolation to find the height (in meters ASL) at the value θ_{sfc} + 0.1°K (i.e., the potential temperature θ at the mixing height). Enter the value found by linear interpolation into the row labeled "MIXING HEIGHT" under the column "HEIGHT (mASL)" and proceed to Step 8.

Step 7 -- Use linear interpolation to find the height above sea level of the mixing height using the pressure at the mixing height (found in Step 5) and the pressure levels on the sounding above and below the mixing height pressure that have both pressure and height values. Enter the height value found into the row "MIXING HEIGHT" under the column marked "HEIGHT, (mASL)" and proceed to Step 8.

Step 8 -- Subtract the elevation of the urban site (mASL) from the height (mASL) of the mixing height. The result is the height of the mixing height in meters above the surface of the city (mAGL). Enter this value into Table A-4.

NOTE; Despite the fact that pressure and height, and potential temperature and height, are not linearly related, linear interpolation does not produce significant errors over the limited ranges used above.

* For example, if the urban surface pressure is 985 mb, and the sounding pressures are: 1005, 1000, 963, 850 mb, etc., 963 mb is the "highest pressure level that is less than the pressure at the urban surface." 850 mb is the "next lowest pressure level" needed in Step 4.

Figure 1. Example of a Step-by-step Mathematical Method for Estimating Mixing Heights



Figure 2. This is a graphical example of the method in Figure 1. The solid line is the temperature sounding and the dashed line is the temperature the surface air (⊗) would have at different pressures (rising without exchanging heat with surrounding air). The air from the surface will rise as long as it is warmer (i.e., less dense) than the surrounding air. The mixing height according to this method is where the air rising from the surface becomes colder (i.e., denser) than the surrounding air and will not rise further.

term for the temperature a parcel of air would have if it were moved vertically, without exchanging heat with the surrounding air, to a pressure of 1000 millibars. The calculated potential temperature of the surface air is compared with the potential temperature calculated for the sounding level nearest to the surface (i.e., layer 1). If the potential temperature at the surface is greater (i.e., warmer) than the potential temperature of layer 1, the surface air will rise through that layer because warm air is less dense than cold air. The potential temperature is computed for the next highest layer of air (i.e., layer 2) and all the layers above it (i.e., layers 3, 4, etc.) until a layer is found that has a potential temperature warmer than the surface potential temperature. When the air rising from the surface enters a layer of air that is warmer (i.e., less dense) than it is, the surface air will stop rising. This altitude is the upper limit of the mixing of air from the surface and is called the mixing height. The height of the top of this well-mixed layer will determine the vertical extent to which pollutants emitted near the earth's surface will be diluted.

The program described herein defines the mixing height as the lowest altitude where the potential temperature is greater than the surface potential temperature. If a sounding has height values for all the pressure levels needed, the program estimates the mixing height by linear interpolation from potential temperature to height. However, not all atmospheric soundings have height values for each pressure level. Where height values are missing, the pressure value at the mixing height is found by linear interpolation of pressure values associated with potential temperature values above and below the mixing height. The elevation of the mixing height is then found by linear interpolation of heights associated with the previously described pressure values.

3.0 APPROXIMATIONS

Since the points used to describe a temperature sounding (i.e., a vertical temperature profile) are connected by straight lines, we can assume that temperature is linear with height between the plotted points. Potential temperature is roughly linear with temperature and therefore no significant error results if potential temperature is taken as being linear with height between the reported sounding levels.

Some error may occur if height values are not given for each pressure level. If a pressure level does not have a height value, temperature (and therefore potential temperature) may not be linear with height between the levels where height values are given. Therefore, this program interpolates from potential temperature to pressure and then from pressure to height. Even though potential temperature is not linear with pressure, the error is not significant over the small pressure intervals often encountered. The estimated pressure at the mixing height is used to find the elevation of the mixing height by linear interpolation using sounding levels that have both pressure and height. While pressure is not linear with height, detailed height and pressure values from soundings taken for the St. Louis Regional Air Pollution Study show that linear interpolation from pressure to height over the interval from 1000 to 850 millibars (about 1500 meters) gives a range of 0 to 40 meters above the actual value. This error will approach zero for low mixing heights and will not be significant for maximum mixing heights, where accuracy of 50 to 100 meters is acceptable for most purposes.

The change of mixing height due to errors in calculated potential temperature cannot be evaluated since the error depends on the rate of change of temperature (and thus potential temperature) with height near the mixing height.

4.0 INPUT DATA FORMAT

To compute the mixing height for the city being modeled, this program needs the following:

1) upper air sounding data (e.g., National Weather Service, local urban radiosonde or helicopter spirals);

2) urban surface temperature and pressure at 0800 LCT* and at the time of maximum temperature;

3) elevation of the urban temperature measurement in meters above sea level, and

4) the climatological daily maximum mixing height value for summer nonprecipitation days (listed for some cities in Reference 1 [page A-3] and for others in Reference 2).

If the pressure data are not available at the elevation of the temperature data, other data can be used if they are adjusted to the elevation of the temperature measurement. Additional information on how to obtain these data is given in

---

* LCT stands for Local Civil Time, which is the prevailing local time for a location. In other words, LCT is Local Standard Time, unless the locality is on Local Daylight Time.

Appendix A of the EKMA Guideline document.[1] Also, Table 1 gives the urban surface and upper air data used in the examples explained below.

The card format to be used to enter the data into the program is in Table 2. The first number on card 1 tells the program that it is estimating the mixing height for either 0800 LCT or the maximum mixing height. The second number on the first card is the climatological daily maximum mixing height value. This value allows the program to check for unusual estimated mixing heights. The urban surface data are contained on the second data card. Sounding data should be entered on the following cards, except for the surface data from the sounding, which was replaced by the urban surface data entered on card 2. In Figure 3, the surface and sounding data used to calculate maximum mixing height is used as an example to explain the input data. The sounding data must be entered in order of decreasing pressure (i.e., increasing height). Sounding levels with a pressure greater than the urban surface pressure or a height less than the urban surface height will be ignored by the program. Additional instructions and cautions are given in the program listing. Significant comments on program operation are marked with an asterisk (*) in the program listing.

Tables 3 and 4 show the input and output for maximum and 0800 LCT mixing height examples, respectively. Users should attempt to replicate the examples illustrated in Tables 3 and 4 using their own computer system prior to using the program for general application.

Table 1. Examples of the Data Needed for This Method

Surface Data

Hour Starting at, LCT	Temperature °C	Pressure, mb
8	23.2	1010.3
9	23.9	1010.7
10	25.8	1010.8
11	27.3	1010.6
12	28.7	1010.3
13	29.3	1010.0
14	30.1	1009.6
15	30.4	1009.2
16	30.8	1008.8
17	31.4	1008.6
18	31.2	1008.5

Sounding Data

1200 GMT Sounding			0000 GMT Sounding		
Pressure (mb)	Height (m ASL)	Temp. (°C)	Pressure (mb)	Height (m ASL)	Temp. (°C)
S 1015*	8*	23.0*	S 1012*	8*	31.0*
M 1000	139	23.0	M 1000	114	30.6
S 967	---	24.4	M 850	1537	16.4
M 850	1550	16.2	S 831	---	15.4
S 827	---	14.2	S 791	---	13.2
S 817	---	13.6	S 778	---	11.8
M 700	3168	4.6	S 760	---	11.2
S 680	---	5.6	M 700	3164	7.0
S 661	---	5.6	S 628	---	1.6
S 608	---	0.4	S 560	---	-1.5
M 500	5860	-8.3	M 500	5860	-7.3
S 491	---	-9.3	M 400	7560	-18.9
S 453	---	-12.7	S 371	---	-21.7
S 438	---	-13.9	M 300	9650	-33.1
M 400	7560	-18.7	S 265	---	-39.9
S 388	---	-20.1	M 250	10900	-42.9
S 349	---	-26.3	S 205	---	-52.9
S 324	---	-29.7	M 200	12370	-53.3
M 300	9640	-33.7	M 150	14190	-61.1
S 267	---	-39.5	S 127	---	-64.9
M 250	10890	-47.7	S 120	---	-61.7
M 200	12370	-51.7	M 100	16690	-63.3
M 150	14190	-60.9	M 70	18900	-58.5
S 148	---	-61.5	M 50	21040	-54.5
			M 30	24350	-49.9
			M 20	27030	-44.7
			S 15	---	-42.1

Note: M = Mandatory Levels and S = Significant Levels

The 0000 GMT Sounding is the following day in GMT.

* The lowest level of the sounding should not be used in the mixing height calculations for EKMA/OZIPP.

Printed copies of National Weather Service sounding data can be ordered by writing to: National Climatic Center, Federal Building, Asheville, North Carolina 28801 or calling (704) 258-2850, extension 683. The cost (as of April 1981) is 60 cents per sounding, with a minimum order of $5.00.

Table 2. Card Input Format

Card Number	Variable Name	Column Number(s)	FORTRAN Format	Data to be Entered
1	MNMX	4	I1	Mixing height to be computed: 0 = 0800 LCT 1 = Maximum
	CLIMPM	5-11	F7.0	Climatological daily maximum mixing height value in meters above ground level
2				URBAN SURFACE DATA
	ELEV	2-9	F8.1	Height in meters above sea level
	PRESS	10-17	F8.1	Pressure in millibars
	TEMP	18-23	F6.1	Temperature in degrees Celsius
3 to last card	ELEV PRESS TEMP	As on card 2, except input is the sounding data without the sounding surface data		

Note: Pressure reduced to sea level should not be used unless the height of the pressure level is at sea level.

- This program was designed to use temperatures to the nearest 0.1°C. Temperatures measured in degrees Fahrenheit should be converted to the nearest 0.1°C. As explained in Section 2.0, pressure can be rounded to the nearest whole millibar and height to the nearest 10 meters.
- Missing values can be entered using a height greater than 90000 or a temperature greater than 900. If a level does not have a pressure value, do not use that level in the input for the program, since the program cannot use a level unless a pressure value is given.

Card Number

1.	1	1700.	
2.	62.0	1008.6	31.4
3.	114.0	1000.0	30.6
4.	1537.0	850.0	16.4
5.	99999.9	831.0	15.4
6.	99999.9	791.0	13.2
7.	99999.9	778.0	11.8
8.	99999.9	760.0	11.2
9.	3164.0	700.0	7.0

Explanation

Card Number

1. "1" indicates that the data is for a maximum mixing height estimation. 1700 meters is the climatological daily maximum mixing height.

2. Urban surface elevation (62.0 meters above sea level), pressure (1008.6 millibars) and temperature (31.4°C).

3-9. Sounding data (elevation, pressure, temperature). 99999.9 = missing height value.

Figure 3. Data Input for Maximum Mixing Height Example

Table 3. Example for Maximum Mixing Height

INPUT:

```
  1  1700.
   62.0  1008.6  31.4
  114.0  1000.0  30.6
 1537.0   850.0  16.4
99999.9   831.0  15.4
99999.9   791.0  13.2
99999.9   778.0  11.8
99999.9   760.0  11.2
 3164.0   700.0   7.0
@EOF
```

OUTPUT:

HEIGHT MASL	PRESSURE MB	TEMP. DEG.C	POTENTIAL TEMP. DEG.K
62.0	1008.6	31.4	303.9
114.0	1000.0	30.6	303.8
1537.0	850.0	16.4	303.4
99999.9	831.0	15.4	304.3
99999.9	791.0	13.2	.0
99999.9	778.0	11.8	.0
99999.9	760.0	11.2	.0
3164.0	700.0	7.0	.0

MAX. MIXING HEIGHT 1613. METERS AGL, 837.3 MILLIBARS.

THE CLIMATOLOGICAL MAXIMUM MIXING HEIGHT VALUE ENTERED WAS 1700. METERS AGL.

Table 4. Example for 0800 LCT Mixing Height

INPUT:

```
   0  1700.
    62.0  1010.3  23.2
   139.0  1000.0  23.0
 99999.9   967.0  24.4
  1550.0   850.0  16.2
 99999.9   827.0  14.2
 99999.9   817.0  13.6
  3168.0   700.0   4.6
@EOF
```

OUTPUT:

HEIGHT	PRESSURE	TEMP.	POTENTIAL TEMP.
MASL	MB	DEG.C	DEG.K
62.0	1010.3	23.2	295.5
139.0	1000.0	23.0	296.2

```
ACCORDING TO THIS METHOD, THE LOWEST LAYER OF THE SOUNDING IS
NOT WELL MIXED.  THIS IMPLIES A MIXING HEIGHT OF ZERO METERS AGL.
THE URBAN MIXING HEIGHT IS GREATER THAN THE    0. METER
MIXING HEIGHT COMPUTED BY THIS METHOD.  250 METERS AGL
SHOULD BE USED FOR THE EKMA 0800 LCT MIXING HEIGHT.

THE CLIMATOLOGICAL MAXIMUM MIXING HEIGHT VALUE ENTERED WAS  1700. METERS AGL.
```

5.0 PROGRAM OUTPUT

The output from the program gives the height of the mixing height in meters above the surface elevation entered on data card 2. Figure 4 explains the output from the input data in Figure 3. When unusual situations occur (e.g., not enough data entered, possibly unrealistic mixing height values), the program informs the user by various messages displayed at the end of the output. Most of these messages direct the user to take some action or to refer to the EKMA Guideline document[1] for further information. These program warnings and why they occur are discussed below. When a problem occurs, the first step should be to check that the data were entered correctly.

OUTPUT -

THE URBAN MIXING HEIGHT IS GREATER THAN THE _____ METER
MIXING HEIGHT COMPUTED BY THIS METHOD. 250 METERS AGL
SHOULD BE USED FOR THE EKMA 0800 LCT MIXING HEIGHT.

EXPLANATION -

- A mixing height less than 250 meters above ground level has been estimated by the program for the 0800 LCT mixing height. Data from St. Louis and Philadelphia temperature soundings taken by helicopter indicate that this program's method of estimating the mixing height underestimates the depth of the early morning urban mixed layer.

OUTPUT

MIXING HEIGHT VALUE MAY BE TOO HIGH.

HEIGHT MASL	PRESSURE MB	TEMP. DEG.C	POTENTIAL TEMP. DEG.K
62.0	1008.6	31.4	303.9
114.0	1000.0	30.6	303.8
1537.0	850.0	16.4	303.4
99999.9	831.0	15.4	304.3
99999.9	791.0	13.2	.0
99999.9	778.0	11.8	.0
99999.9	760.0	11.2	.0
3164.0	700.0	7.0	.0

MAX. MIXING HEIGHT 1613. METERS AGL, 837.3 MILLIBARS.

THE CLIMATOLOGICAL MAXIMUM MIXING HEIGHT VALUE ENTERED WAS 1700. METERS AGL.

Explanation -

- If height values are input in meters above ground level (mAGL), the output in the "height" column will be in mAGL instead of meters above sea level (mASL) as labeled. The mixing height value is in meters above the urban surface elevation entered in columsn 2 through 9 of card 2.

- Values of height, pressure and temperature in the output are the same as the input values. The values of potential temperature in degrees Kelvin by the program where needed.

- The mixing height is between 850 and 831 millibars (MB) at a potential temperature of 304.0°K. The height values used for interpolation are at 850 and 700 MB and the mixing height is estimated to be at 1613. meters above the urban surface elevation of 62 mASL entered on card 2 of the input. The estimated mixing height is significant to the nearest 10 meters at best.

Figure 4. Detailed Example of Output for Maximum Mixing Height Estimation from the Input Shown in Figure 3.

EXPLANATION

- For 0800 LCT mixing height - estimated value is greater than 500 meters above ground level. An incorrect value for urban surface temperature may have been entered.

- For the maximum mixing height - estimated value is more than twice the climatological value entered on card 1. An incorrect value may have been entered for the urban surface temperature or the sounding may be unrepresentative of the region being modeled.

OUTPUT

MIXING HEIGHT IS LOW FOR A MAXIMUM MIXING HEIGHT.

EXPLANATION

- The estimated maximum mixing height is less than one-third the climatological value entered on card 1 of the data for the program or is less than 250 meters above ground level. The sounding may be from a location that is much warmer than the urban area being modeled. If that is true, another more representative site should be found.

OUTPUT

SEE **USE OF CITY-SPECIFIC EKMA FOR OZONE SIPS**

FOR ALTERNATIVE PROCEDURES

EXPLANATION

- This is a followup for other messages when possible solutions are outlined in the EKMA Guideline document.[1]

OUTPUT

ACCORDING TO THIS METHOD, THE LOWEST LAYER OF THE SOUNDING IS NOT WELL MIXED. THIS IMPLIES A MIXING HEIGHT OF ZERO METERS AGL.

EXPLANATION

- Even though this method gives a mixing height of zero meters above ground level, either a low morning mixing height exists or a problem has been caused by the data used for the maximum mixing height values. See other statements listed by the program output for possible solutions. The program prints the statement because the potential temperature of the surface air is lower (colder) than the potential temperature of the layer above it.

OUTPUT

DATA FINISHED BEFORE MIXING HEIGHT FOUND - POSSIBLE BAD SURFACE DATA.

EXPLANATION

- Check to see if the correct urban surface temperature value was entered on card 2. Otherwise the mixing height is above the highest sounding level entered into the program. Also check to see if the sounding used is representative of the urban region.

MIXING HEIGHT WAS FOUND BUT HEIGHT VALUE ABOVE MIXING HEIGHT LEVEL OF _____ MILLIBARS IS NEEDED FOR INTERPOLATION.

EXPLANATION

- To do linear interpolation from pressure to height, the program needs height values above and below the estimated mixing height. If no upper height value exists, use a height value from another nearby sounding on the same day or a height value from the U.S. Standard Atmosphere.

6.0 REFERENCES

1. U.S. Environmental Protection Agency, Guideline for Use of City-specific EKMA in Preparing Ozone SIPs, EPA-450/4-80-027 (March 1981).

2. G. C. Holzworth, Mixing Heights, Wind Speeds, and Potential for Urban Air Pollution Throughout the Contiguous United States, AP-101, U.S. Environmental Protection Agency, Research Triangle Park, North Carolina (January 1972).

7.0 PROGRAM LISTING

```
TDS*MXHGT(1).MXHGT/EKMA
C     *******************************************************************MHX00100
C     ***COMPUTERIZED METHOD FOR ESTIMATING ATMOSPHERIC MIXING HEIGHTS***MHX00200
C     **********VERSION FOR USE WITH EKMA/OZIPP, MARCH 25, 1981**********MHX00300
C     *******************************************************************MHX00400
C   * NOTE: STARRED COMMENTS (*) MUST BE FOLLOWED FOR THIS PROGRAM       MHX00500
C     TO WORK PROPERLY.                                                 MHX00600
C                                                                       MHX00700
C    THIS PROGRAM FINDS THE MIXING HEIGHT USING URBAN SURFACE AND       MHX00800
C    RADIOSONDE SOUNDING DATA.  THE PROGRAM WAS WRITTEN BY BOB KELLY    MHX00900
C    OF EPA'S AIR MANAGEMENT TECHNOLOGY BRANCH AND GIVES THE MIXING     MHX01000
C    HEIGHT BY FINDING THE LOWEST HEIGHT ON THE SOUNDING WHERE THE      MHX01100
C    POTENTIAL TEMPERATURE IS GREATER THAN THE POTENTIAL TEPERATURE     MHX01200
C    AT THE SURFACE.                                                    MHX01300
      DIMENSION ELEV(50),PRESS(50),TEMP(50),PT(50)                      MHX01400
      REAL MIXHT                                                        MHX01500
C                                                                       MHX01600
C    UNIT NUMBERS FOR INPUT AND OUTPUT FOR THIS PROGRAM CAN BE CHANGED  MHX01700
C    BY THE USER IF UNIT NUMBERS OTHER THAN THE '5' (FOR CARD READER)   MHX01800
C    AND THE '6' (LINE PRINTER) USED IN THIS PROGRAM ARE NEEDED.        MHX01900
C    THIS CAN BE DONE BY CHANGING THE VALUES FOR 'IN' AND 'OUT' ON THE  MHX02000
C    DATA CARD BELOW TO THE INPUT AND OUTPUT UNIT NUMBERS NEEDED.       MHX02100
      INTEGER IN,OUT                                                    MHX02200
      DATA IN/5/,OUT/6/                                                 MHX02300
      NN = 0                                                            MHX02400
      READ(IN,250)  MNMX, CLIMPM                                        MHX02500
  250 FORMAT (3X, I1, F7.0)                                             MHX02600
C   *'MNMX'- FOR 0800 LCT ENTER '0', FOR MAXIMUM MIXING HEIGHT ENTER '1'.MHX02700
C   *'CLIMPM' IS THE CLIMATOLOGICAL MAXIMUM MIXING HEIGHT               MHX02800
C     IN METERS ABOVE GROUND LEVEL.                                     MHX02900
C                                                                       MHX03000
C   *IF MORE THAN 50 LEVELS ARE TO BE USED, THE SIZE OF THE ARRAYS      MHX03100
C     IN THE DIMENSION STATEMENT (ABOVE) SHOULD BE INCREASED TO THE     MHX03200
C     MAXIMUM NUMBER OF LEVELS TO BE USED.                              MHX03300
C                                                                       MHX03400
C   *THE URBAN SURFACE DATA MUST BE INPUT FIRST IN PLACE OF THE SOUNDING MHX03500
C     SURFACE DATA AND MUST HAVE OBSERVED VALUES FOR ELEVATION, PRESSURE MHX03600
C     AND TEMPERATURE.  THE SOUNDING DATA MUST BE ENTERED IN ORDER OF   MHX03700
C     INCREASING HEIGHT (I.E. DECREASING PRESSURE).                     MHX03800
C                                                                       MHX03900
C   *ALL LEVELS ENTERED MUST HAVE PRESSURE VALUES.                      MHX04000
C                                                                       MHX04100
C   *THE LAST LEVEL OF THE SOUNDING DATA FOR THIS PROGRAM MUST          MHX04200
C     HAVE OBSERVED VALUES FOR HEIGHT, PRESSURE AND TEMPERATURE.        MHX04300
      READ(IN,150) ELEV(1), PRESS(1), TEMP(1)                           MHX04400
  150 FORMAT (2F8.1, F6.1)                                              MHX04500
C   *       'ELEV'  = HEIGHT ABOVE SEA LEVEL IN METERS.                 MHX04600
C           'PRESS' = PRESSURE IN MILLIBARS.                            MHX04700
C           'TEMP'  = TEMPERATURE TO THE NEAREST 0.1 DEGREES CELSIUS.   MHX04800
C   *FOR MISSING VALUES USE THE FOLLOWING:                              MHX04900
C       FOR HEIGHT      ('ELEV')  USE '99999.9',                        MHX05000
C       FOR PRESSURE    ('PRESS') MISSING VALUES ARE NOT ALLOWED,       MHX05100
C       FOR TEMPERATURE ('TEMP')  USE   '999.9'.                        MHX05200
      PT(1) = (TEMP(1)+273.2)*((PRESS(1)/1000.)** -0.286)               MHX05300
      PT(1) = (AINT((PT(1)+0.05)*10.))/10.                              MHX05400
C    'PT' IS THE POTENTIAL TEMPERATURE TO THE NEAREST 0.1 DEGREES       MHX05500
C    KELVIN FROM VALUES OF TEMPERATURE AND PRESSURE AT LEVEL (L).       MHX05600
```

```
C                                                                        MHX05700
C   THE URBAN SURFACE DATA IS AT L = 1.                                  MHX05800
      L = 2                                                              MHX05900
   20 READ(IN,150, END=55) E,P,T                                         MHX06000
      IF (P .GE. PRESS(1)) GO TO 20                                      MHX06100
      IF (E .LE. ELEV(1)) GO TO 20                                       MHX06200
      ELEV(L) = E                                                        MHX06300
      PRESS(L) = P                                                       MHX06400
      TEMP(L) = T                                                        MHX06500
      IF (T .GT. 900.) GO TO 18                                          MHX06600
      GO TO 17                                                           MHX06700
   18 IF (E .GT. 90000.) GO TO 20                                        MHX06800
      GO TO 19                                                           MHX06900
   17 PT(L) = (TEMP(L)+273.2)*((PRESS(L)/1000.)** -0.286)                MHX07000
      PT(L) = (AINT((PT(L)+0.05)*10.0))/10.                              MHX07100
      IF (PT(L) .GT. PT(1)) GO TO 30                                     MHX07200
   19 L = L+1                                                            MHX07300
      GO TO 20                                                           MHX07400
   30 K = L                                                              MHX07500
      IF (L .EQ. 2) GO TO 95                                             MHX07600
      N = L-1                                                            MHX07700
   33 IF (TEMP(N) .GT. 900.) GO TO 34                                    MHX07800
      GO TO 41                                                           MHX07900
   34 N = N-1                                                            MHX08000
      GO TO 33                                                           MHX08100
   41 PRESMH = ((PRESS(L)-PRESS(N))*((PT(1)+0.1)-PT(L))/                 MHX08200
     7 (PT(L)-PT(N)))+PRESS(L)                                           MHX08300
      PRESMH = (AINT((PRESMH+0.05)*10.))/10.                             MHX08400
C   'PRESMH' IS THE PRESSURE AT THE MIXING HEIGHT                        MHX08500
C    TO THE NEAREST 0.1 MILLIBARS.                                       MHX08600
   50 IF (ELEV(K) .GT. 90000) GO TO 40                                   MHX08700
      GO TO 51                                                           MHX08800
   40 K = K+1                                                            MHX08900
      READ(IN,150, END=65) ELEV(K), PRESS(K), TEMP(K)                    MHX09000
C   THIS READ STATEMENT IS USED TO READ ADDITIONAL SOUNDING LEVELS       MHX09100
C   TO FIND AN UPPER VALUE FOR INTERPOLATION FROM PRESSURE TO HEIGHT.    MHX09200
      GO TO 50                                                           MHX09300
   51 J = L-1                                                            MHX09400
   70 IF (ELEV(J) .GT. 90000) GO TO 60                                   MHX09500
      GO TO 80                                                           MHX09600
   60 J = J-1                                                            MHX09700
      GO TO 70                                                           MHX09800
   80 ELEVMH = ((ELEV(K)-ELEV(J))*(PRESMH-PRESS(K))/                     MHX09900
     7 (PRESS(K)-PRESS(J)))+ELEV(K)                                      MHX10000
C   'ELEVMH' IS THE MIXING HEIGHT                                        MHX10100
C   IN METERS ABOVE SEA LEVEL (MASL).                                    MHX10200
      MIXHT = ELEVMH-ELEV(1)                                             MHX10300
      MIXHT = AINT(MIXHT+0.5)                                            MHX10400
C   'MIXHT' IS THE MIXING HEIGHT IN METERS ABOVE GROUND LEVEL            MHX10500
C   (MAGL) ROUNDED TO THE NEAREST METER.                                 MHX10600
C                                                                        MHX10700
C   THE FOLLOWING SECTION GIVES OUTPUT                                   MHX10800
C   AND CHECKS FOR EXTREME VALUES.                                       MHX10900
C                                                                        MHX11000
   95 WRITE(OUT,1003)                                                    MHX11100
 1003 FORMAT (/ 29X 'POTENTIAL'/,                                        MHX11200
     7 '    HEIGHT  PRESSURE   TEMP.    TEMP.'/,                         MHX11300
```

```
      7 '     MASL         MB      DEG.C     DEG.K')                        MHX11400
        DO 1001 M = 1,K                                                   MHX11500
        WRITE(OUT,1000) ELEV(M), PRESS(M), TEMP(M), PT(M)                 MHX11600
 1000 FORMAT (4F9.1)                                                      MHX11700
 1001 CONTINUE                                                            MHX11800
        IF (NN .EQ. 1) GO TO 85                                           MHX11900
        IF (L .EQ. 2) GO TO 46                                            MHX12000
        IF (MNMX .EQ. 1) GO TO 15                                         MHX12100
        WRITE(OUT,100) MIXHT, PRESMH                                      MHX12200
  100 FORMAT (//'   A.M. MIXING HEIGHT',F7.0, ' METERS AGL,',             MHX12300
     7  ,F7.1, ' MILLIBARS.')                                             MHX12400
        IF (MIXHT .LT. 250.) GO TO 25                                     MHX12500
        IF (MIXHT .GT. 500.) GO TO 35                                     MHX12600
        GO TO 85                                                          MHX12700
   25 WRITE(OUT,300) MIXHT                                                MHX12800
  300 FORMAT ( '   THE URBAN MIXING HEIGHT IS GREATER THAN THE',F5.0,     MHX12900
     7 ' METER'/'   MIXING HEIGHT COMPUTED BY THIS METHOD.  250 METERS',MHX13000
     7 ' AGL'/'   SHOULD BE USED FOR THE EKMA 0800 LCT MIXING HEIGHT.') MHX13100
        GO TO 85                                                          MHX13200
   15 XCLIM = CLIMPM * 2.                                                 MHX13300
        WRITE(OUT,200) MIXHT, PRESMH                                      MHX13400
  200 FORMAT (//'   MAX. MIXING HEIGHT',F7.0, ' METERS AGL,',             MHX13500
     7  ,F7.1, ' MILLIBARS.')                                             MHX13600
   49 IF (MIXHT .GT. XCLIM) GO TO 35                                      MHX13700
        IF (MIXHT .LE. 250) GO TO 45                                      MHX13800
        YCLIM = CLIMPM/3.                                                 MHX13900
        IF (MIXHT .LE. YCLIM) GO TO 45                                    MHX14000
        GO TO 85                                                          MHX14100
   35 WRITE(OUT,400)                                                      MHX14200
  400 FORMAT ( '   MIXING HEIGHT VALUE MAY BE TOO HIGH.')                 MHX14300
        GO TO 86                                                          MHX14400
   45 WRITE(OUT,500)                                                      MHX14500
  500 FORMAT ( '   MIXING HEIGHT IS LOW FOR',                             MHX14600
     7 ' A MAXIMUM MIXING HEIGHT.')                                       MHX14700
   86 WRITE(OUT,350)                                                      MHX14800
  350 FORMAT ( '   SEE **USE OF CITY-SPECIFIC EKMA FOR OZONE SIPS**'/,    MHX14900
     7 '   FOR ALTERNATIVE PROCEDURES.')                                  MHX15000
        GO TO 85                                                          MHX15100
   46 WRITE(OUT,450)                                                      MHX15200
  450 FORMAT (/'   ACCORDING TO THIS METHOD, THE LOWEST LAYER OF THE',    MHX15300
     7 ' SOUNDING IS'/'   NOT WELL MIXED.  THIS IMPLIES A MIXING',        MHX15400
     7 ' HEIGHT OF ZERO METERS AGL.')                                     MHX15500
        IF (MNMX .EQ. 1) GO TO 45                                         MHX15600
        GO TO 25                                                          MHX15700
   55 WRITE(OUT,600)                                                      MHX15800
  600 FORMAT ( '   DATA FINISHED BEFORE MIXING HEIGHT FOUND-',            MHX15900
     7 'POSSIBLE BAD SURFACE DATA.')                                      MHX16000
        K = L-1                                                           MHX16100
        NN = 1                                                            MHX16200
        GO TO 95                                                          MHX16300
   65 WRITE(OUT,700) PRESMH                                               MHX16400
  700 FORMAT ( '   MIXING HEIGHT WAS FOUND BUT HEIGHT VALUE ABOVE',       MHX16500
     7 ' MIXING HEIGHT LEVEL'/'   OF',F7.1,' MILLIBARS IS NEEDED FOR',    MHX16600
     7 ' INTERPOLATION.')                                                 MHX16700
        K = K-1                                                           MHX16800
        NN = 1                                                            MHX16900
        GO TO 95                                                          MHX17000
   85 WRITE(OUT,800) CLIMPM                                               MHX17100
  800 FORMAT ( /'   THE CLIMATOLOGICAL MAXIMUM MIXING HEIGHT VALUE',      MHX17200
     7 ' ENTERED WAS', F7.0, ' METERS AGL.')                              MHX17300
        STOP                                                              MHX17400
        END                                                               MHX17500
```

TECHNICAL REPORT DATA

(Please read Instructions on the reverse before completing)

1. REPORT NO. EPA-450/4-81-022	2.	3. RECIPIENT'S ACCESSION NO.
4. TITLE AND SUBTITLE User's Manual for Mixing Height Computer Program		5. REPORT DATE May 1981
		6. PERFORMING ORGANIZATION CODE
7. AUTHOR(S) Robert F. Kelly		8. PERFORMING ORGANIZATION REPORT NO.
9. PERFORMING ORGANIZATION NAME AND ADDRESS U.S. Environmental Protection Agency Office of Air Quality Planning and Standards Monitoring and Data Analysis Division/AMTB (MD-14) Research Triangle Park, North Carolina 27711		10. PROGRAM ELEMENT NO.
		11. CONTRACT/GRANT NO.
12. SPONSORING AGENCY NAME AND ADDRESS Same		13. TYPE OF REPORT AND PERIOD COVERED
		14. SPONSORING AGENCY CODE

15. SUPPLEMENTARY NOTES

Supplements EPA-450/4-80-027, "Guideline for Use of City-specific EKMA in Preparing Ozone SIPs."

16. ABSTRACT

A FORTRAN-language computer program has been developed to estimate mixing height values for use in the Empirical Kinetic Modeling Approach/Ozone Isopleth Plotting Package (EKMA/OZIPP). This program uses temperature, pressure and height values measured at a surface site, and from atmospheric vertical profiles (e.g., radiosondes). The manual contains instructions on how to use the program, including a description of the data needed, how to format the data, and an explanation of the output from the program. Examples of input and output are also included.

17. KEY WORDS AND DOCUMENT ANALYSIS

a. DESCRIPTORS	b. IDENTIFIERS/OPEN ENDED TERMS	c. COSATI Field/Group
Mixing Height Computer Program EKMA OZIPP		

18. DISTRIBUTION STATEMENT Unlimited	19. SECURITY CLASS *(This Report)* Unclassified	21. NO. OF PAGES 28
	20. SECURITY CLASS *(This page)* Unclassified	22. PRICE

EPA Form 2220-1 (Rev. 4-77) PREVIOUS EDITION IS OBSOLETE